



Tech Info Library

Applesoft: Literal Input Routine

This program demonstrates a routine that takes anything you type and enters it into a string. It's very important that `IN$ = "X"` is the first line of the program.

```
100 LET IN$ = "X"
110 TEXT : HOME
120 REM ** THE FIRST VARIABLE DEFINED MUST BE A STRING
130 REM ** THIS STRING WILL RECEIVE INPUT FROM THE CALL
140 REM ** THIS POKES THE INPUT SIMULATOR ROUTINE INTO MEMORY
200 FOR J = 768 TO 790
210 READ I : POKE J,I
220 NEXT J
230 DATA 162,0,32,117,253,160,2
240 DATA 138,145,105,200,169,0
250 DATA 145,105,200,169,2,145
260 DATA 105,76,57,213
300 REM ** NOW TO USE IT!
310 PRINT "TYPE IN ANY CHARACTERS YOU WISH:" : PRINT
320 CALL 768:IN$ = MID$(IN$,1)
330 REM ** THIS IS AN "INPUT IN$" BUT IGNORES ",", " AND ":"
400 PRINT : PRINT "AND HERE'S WHAT YOU TYPED IN:"
420 PRINT : PRINT IN$
430 PRINT : PRINT "NOTE THAT EVEN QUOTES, COMMAS AND"
450 PRINT "COLONS GET THROUGH UNSCATHED."
460 PRINT : PRINT "NOW LET'S WRITE IT TO THE DISK"
470 PRINT CHR$(4);"OPEN TEMP"
480 PRINT CHR$(4);"WRITE TEMP"
490 PRINT IN$
500 PRINT CHR$(4);"CLOSE"
510 PRINT : PRINT "AND READ IT BACK IN..."
520 LET IN$ = " "
530 PRINT CHR$(4);"OPEN TEMP"
540 PRINT CHR$(4);"READ TEMP"
550 CALL 768:IN$ = MID$(IN$,1)
560 PRINT CHR$(4);"CLOSE"
570 PRINT : PRINT IN$
580 PRINT : PRINT "TA-DAA!!": END
```

Apple Tech Notes

Keywords: <None>

=====

This information is from the Apple Technical Information Library.

19960215 11:05:19.00

Tech Info Library Article Number: 102